



InterSystems IRIS での DataMove の使用法

Version 2023.1
2024-01-02

InterSystems IRIS での DataMove の使用法
InterSystems IRIS Data Platform Version 2023.1 2024-01-02
Copyright © 2024 InterSystems Corporation
All rights reserved.

InterSystems®, HealthShare Care Community®, HealthShare Unified Care Record®, IntegratedML®, InterSystems Caché®, InterSystems Ensemble®, InterSystems HealthShare®, InterSystems IRIS®, および TrakCare は、InterSystems Corporation の登録商標です。HealthShare® CMS Solution Pack™ HealthShare® Health Connect Cloud™, InterSystems IRIS for Health™, InterSystems Supply Chain Orchestrator™, および InterSystems TotalView™ For Asset Management は、InterSystems Corporation の商標です。TrakCare は、オーストラリアおよび EU における登録商標です。

ここで使われている他の全てのブランドまたは製品名は、各社および各組織の商標または登録商標です。

このドキュメントは、インターシステムズ社(住所: One Memorial Drive, Cambridge, MA 02142)あるいはその子会社が所有する企業秘密および秘密情報を含んでおり、インターシステムズ社の製品を稼動および維持するためにのみ提供される。この発行物のいかなる部分も他の目的のために使用してはならない。また、インターシステムズ社の書面による事前の同意がない限り、本発行物を、いかなる形式、いかなる手段で、その全てまたは一部を、再発行、複製、開示、送付、検索可能なシステムへの保存、あるいは人またはコンピュータ言語への翻訳はしてはならない。

かかるプログラムと関連ドキュメントについて書かれているインターシステムズ社の標準ライセンス契約に記載されている範囲を除き、ここに記載された本ドキュメントとソフトウェアプログラムの複製、使用、廃棄は禁じられている。インターシステムズ社は、ソフトウェアライセンス契約に記載されている事項以外にかかるソフトウェアプログラムに関する説明と保証をするものではない。さらに、かかるソフトウェアに関する、あるいはかかるソフトウェアの使用から起こるいかなる損失、損害に対するインターシステムズ社の責任は、ソフトウェアライセンス契約にある事項に制限される。

前述は、そのコンピュータソフトウェアの使用およびそれによって起こるインターシステムズ社の責任の範囲、制限に関する一般的な概略である。完全な参照情報は、インターシステムズ社の標準ライセンス契約に記載され、そのコピーは要望によって入手することができる。

インターシステムズ社は、本ドキュメントにある誤りに対する責任を放棄する。また、インターシステムズ社は、独自の裁量にて事前通知なしに、本ドキュメントに記載された製品および実行に対する代替と変更を行う権利を有する。

インターシステムズ社の製品に関するサポートやご質問は、以下にお問い合わせください:

InterSystems Worldwide Response Center (WRC)
Tel: +1-617-621-0700
Tel: +44 (0) 844 854 2917
Email: support@InterSystems.com

目次

InterSystems IRIS での DataMove の使用法	1
1 DataMove の概要	1
2 DataMove の制限事項	1
3 DataMove ワークフロー	2
4 ミラー環境または ECP 環境での DataMove	2
5 DataMove API	3
5.1 DataMove ネームスペース・マッピングの作成と編集	3
5.2 DataMove の生成	4
5.3 DataMove の開始	5
5.4 DataMove の監視	6
5.5 ^DATAMOVE ユーティリティ	7
5.6 マッピング変更の有効化と DataMove の完了	12
5.7 移動元のグローバルの削除と DataMove の完了	13
5.8 その他の API 呼び出し	14
6 DataMove の状態	16
7 DataMove の例	17
8 非推奨の呼び出しと State	18
 テーブル一覧	
テーブル 1: DataMove の State のリスト	16
テーブル 2: 非推奨および新規の DataMove の呼び出しのリスト	19
テーブル 3: 非推奨および新規の DataMove の State のリスト	19

InterSystems IRIS での DataMove の使用法

このドキュメントでは、DataMove プロセスを使用して、InterSystems IRIS® データ・プラットフォームのネームスペースに関連付けられている既存のデータを新しい場所に移動する方法を説明します。

重要 DataMove は、実働環境で使用する前に、特定のネームスペースおよびデータベースを使用してテスト環境でテストしておくことを強くお勧めします。

注釈 このドキュメントに記載されている 1 行のコードは、ページ・サイズやフォント・サイズによっては折り返されて複数行にまたがっている場合があります。

1 DataMove の概要

DataMove プロセスでは、以下の操作を実行することで、InterSystems IRIS ネームスペースに関連付けられている既存のデータを別のデータベースに移動できます。

- ・ ネームスペースに新しいマッピングを作成する。
- ・ マッピングの変更を分析し、移動する必要があるグローバルおよびグローバル添え字を計算する。
- ・ データを新しい 1 つ以上のデータベースにコピーする。
- ・ マッピングの変更を有効にする。

例えば、DataMove を使用すると、グローバルの一部または全部をネームスペースの既定のグローバル・データベースから別のデータベースに移動できます。DataMove プロセスは、ネームスペースのグローバルをルーチンとは別のデータベースに移動する、複数のデータベースにまたがるデータを分割する、アプリケーションの進化に伴って更新された設計上の決定事項に基づいて新しい場所にデータを移動するなどの操作に使用できます。DataMove を使用すると、アプリケーションでデータをアクティブに使用しながら、同じデータを移動し、マッピングを変更できます。

マッピングの詳細は、“システム管理ガイド”の“[ネームスペースへのグローバル、ルーチン、およびパッケージ・マッピングの追加](#)”を参照してください。

非推奨になっている API とその代替機能については、以下の“[非推奨の呼び出しと State](#)”を参照してください。

2 DataMove の制限事項

DataMove プロセスには以下の制限事項が適用されます。

- ・ DataMove は、同一インスタンス上のデータベース間でデータを移動するよう設計されています。ミラーリングされたシステムおよび ECP クライアントが接続されたシステムには、操作上の特別な注意が必要です。
- ・ DataMove は、拡張グローバル参照を使用するアプリケーションでは使用できません。アプリケーションでマッピングと拡張参照が混在している場合、データ統合は保証されません。
- ・ システムでジャーナリングが有効になっている必要があります。
- ・ ジャーナル・エラー発生時にフリーズするスイッチが設定されている必要があります。
- ・ データを移動できるのは、ローカル・データベース間およびミラーリングされたデータベース間のみです。

- ・ ミラーリングされたデータベースを含む DataMove では、プライマリ・ノードでのみ DataMove を実行できます。
- ・ DataMove がプライマリ・ミラーで実行されているときに、フェイルオーバーまたは DR ノードがプライマリになった場合、DataMove は停止されます。DataMove が実行されているノードが再びプライマリになった場合、DataMove は停止した箇所から再開されます。

拡張グローバル参照の詳細は、“ObjectScript の使用法”の“[ネームスペース](#)”を参照してください。

3 DataMove ワークフロー

DataMove ワークフローは、以下のフェーズで構成されています。

1. ネームスペースのマッピングに対する変更が一時ストレージ領域に保存されます。
2. 一連のデータの移動がマッピングから生成されます。
3. データの移動が、指定されたグローバルおよびデータベースに対して検証され、移動先のデータベースに十分なディスク空き容量があることが確認されます。何らかの問題が検出された場合、ユーザはそれを修正して、ワークフローを再開できます。
4. DataMove が開始されると、複数のバックグラウンド・ジョブが既存のデータを新しい場所にコピーします。データがコピーされると、他のバックグラウンド・ジョブがジャーナル・ファイル进行处理して、コピーされたデータに変更を適用します。
5. データのコピーがすべて完了し、ジャーナルが適用されると、ネームスペースで新しいマッピングが有効化されます。
6. ネームスペースへの変更が正常に有効化された後は、新しい場所にコピーされた古いソース・データを削除できます。

DataMove では、すべての処理のログ・ファイル **DataMoveName.log** が **/mgr** ディレクトリに保持されます。また、以前のすべての操作のレコードが **/mgr** ディレクトリの **DataMove.log** ファイルにあります。

4 ミラー環境または ECP 環境での DataMove

ECP 環境またはミラー環境で DataMove を使用する場合、新しいネームスペース・マッピングが ECP クライアントおよびミラーで更新されていることを確認してください。ミラー環境の場合に、新規作成したミラーリングされたデータベースにデータを移動するには、これらのミラーリングされたデータベースをバックアップ・ミラーと非同期ミラーにも作成する必要があります。

1. DataMove を作成し、その DataMove で新しいマッピングを有効化する準備が整うまで DataMove を実行します。
2. フェイルオーバー・ミラーを使用している場合は、有効化する前にバックアップを DR ミラーに降格させる必要があります。
3. ECP クライアントを使用している場合は、有効化する前に、そのクライアント上のアプリケーションを停止するか、ECP システムをダウン状態にする必要があります。
4. 非同期ミラーを使用している場合は、有効化する前に、そこにあるデータにアクセスするすべてのアプリケーションを停止するか、そこに直接接続しているすべての ECP クライアントを停止する必要があります。
5. この時点で、新しい DataMove マッピングを有効化できます。

6. 非同期ミラー上のマッピングを、適用したばかりの新しい DataMove マッピングに一致するよう更新します。これには、以前のバックアップ・フェイルオーバー・ミラーも含まれます。
7. 以前のバックアップ・ミラーを DR ミラーからフェイルオーバー・ミラーに昇格させます。
8. ECP クライアントを、適用したばかりの新しい DataMove マッピングで更新します。
9. 非同期ミラーと ECP クライアントでのアプリケーションの再起動を許可します。

5 DataMove API

このセクションでは、DataMove ワークフローの各フェーズで必要な API 呼び出しについて詳しく説明します。

以下のガイドラインをメモしておいてください。

- ・ 特定の要件に合わせて、必要なスクリプトやユーザ・インタフェースを指定する必要があります。
- ・ 必要なマクロを利用するには、コードに `%syDataMove` ファイルをインクルードする必要があります。
- ・ コードでは、次の API 呼び出しに進む前に、各メソッドから返されるステータスを確認する必要があります。
- ・ 移動先データベースが存在しない場合は、DataMove API を呼び出す前に作成する必要があります。
- ・ ワークフローは、`%SYS` ネームスペースで実行する必要があります。

重要 DataMove を実行する前に、システムの現在のフル・バックアップを作成しておくことをお勧めします。

5.1 DataMove ネームスペース・マッピングの作成と編集

5.1.1 DataMove.API.MapInitialize(Name as %String, Namespaces As %String) As %Status

新しい一連のマッピング編集用の一時ストレージ領域を初期化します。

引数：

- ・ Name は、DataMove の名前です。
- ・ Namespaces は、DataMove を実行するネームスペースのコンマ区切りリストです。

このメソッドは、あらゆる編集を加える前に呼び出す必要があり、指定したネームスペースに対してのみ有効です。複数のネームスペースで、同じ DataMove の一部としてデータを移動できます。

注意 このメソッドを呼び出すことによって一時ストレージ領域内の既存の編集が削除されます。一時ストレージ領域は、DataMove API による変更および管理ポータルの [グローバルマッピング] ページを使用する変更を含め、すべてのネームスペースでのすべてのマッピング変更に使用されます (“システム管理ガイド” の “InterSystems IRIS の構成” の章にある “[グローバル・マッピング](#)” を参照してください)。同時マッピング変更によってエラーを生じさせないために、MapInitialize() を呼び出す際はどのネームスペースに対してもマッピング変更が進行していないこと、および DataMove プロセスが完了するまで他のマッピング変更が開始されないことを確認する必要があります。

例：

```
Set Namespace = "SALES"
Set Status = ##Class(DataMove.API).MapInitialize("DMName",Namespace)
If $$$ISOK(Status) Write !,$$SYSTEM.Status.GetErrorText(Status)
```

例：

```
Set Namespaces = "SALES,PROSPECTS"
Set Status=##Class(DataMove.API).MapInitialize("DMName",Namespaces)
If $$$ISOK(Status) Write !,$SYSTEM.Status.GetErrorText(Status)
```

5.1.2 DataMove.API.MapGlobalsCreate(Name as %String, Namespace As %String, GblName As %String, ByRef Properties As %String) As %Status

一時ストレージ領域に、このネームスペースの新しいグローバル・マッピングを作成します。このメソッドは、この DataMove に含める予定のマッピングの数に応じて、1 回以上呼び出すことができます。

引数：

- ・ Name は、DataMove の名前です。
- ・ Namespace は、DataMove を実行するネームスペースです。
- ・ GblName は、特定のデータベースにマッピングするグローバルの名前です。
- ・ Properties は、このマッピングに必要なプロパティ (特にこのマッピングで使用するデータベースの名前) の配列です。

Name 引数を “A” に設定すると、このマッピングをグローバル ^A 全体に適用することが指定されます。この引数を “A(5):A(10)” に設定すると、このマッピングを ^A(5) から ^A(10) まで (ただし ^A(10) は含まない) の添え字付きのグローバル範囲に適用することが指定されます。

Properties 引数を、Properties("Database") が “USER2” に設定される配列 Properties に設定すると、グローバル (またはグローバル範囲) がデータベース USER2 にマッピングされることが指定されます。

マッピングを変更または削除するには、それぞれ MapGlobalsModify() メソッド、API.MapGlobalsDelete() メソッドを使用できます。詳細は、“クラス・リファレンス”を参照してください。

例：

```
Set Properties("Database")="DSTDB"

;Move ^X(100) up to but not including ^X(200) to database DSTDB
Set Status = ##Class(DataMove.API).MapGlobalsCreate("DMName","SALES","X(100):(200)","Properties)

;Move the entire ^INFO global to database DSTDB
Set Status = ##Class(DataMove.API).MapGlobalsCreate("DMName","SALES","INFO","Properties)

;Move all Data from the first subscript in ^BILLING
;(including the node ^BILLING itself) up to but not including ^BILLING(100)
;to database DSTDB
Set Status = ##Class(DataMove.API).MapGlobalsCreate("DMName","SALES","BILLING(BEGIN):(100)","Properties)

;Move the entire global ^PROSPECT
;from its currently mapped database to DSTDB
Set Status = ##Class(DataMove.API).MapGlobalsModify("DMName","SALES","PROSPECT","Properties)

;Move the entire global ^ORDERS from its currently mapped database
;back to the default database for the namespace SALES
Set Status=##Class(DataMove.API).MapGlobalsDelete("DMName","SALES","ORDERS")
```

グローバル範囲を定義する方法の詳細は、“システム管理ガイド”の“[グローバル・マッピング](#)”を参照してください。

5.2 DataMove の生成

5.2.1 DataMove.API.Generate(Name As %String, ByRef Properties As %String, ByRef Warnings As %String, ByRef Errors As %String) As Status

一時ストレージ領域のマッピング編集に基づいて新しい DataMove を作成します。

引数：

- ・ Name は、作成する DataMove の名前です。

- ・ Properties は、DataMove の作成に使用するオプションのプロパティの配列です。
- ・ Warnings は、DataMove の実行を阻止することはない競合を含んで返される配列です。
- ・ Errors は、DataMove の実行を阻止する競合を含んで返される配列です。

Properties が、Properties 引数として渡される配列の場合、次のようになります。

- ・ Properties("MaxMBPerMin") では、DataMove 処理で移動先データベースに移動できる 1 分あたりの最大 MB 数をオプションで指定します。これを 0 に設定すると、DataMove は可能な限り高速で実行されます。これを渡さない場合は、システムの既定値が使用されます。
- ・ Properties("Description") では、オプションで実行対象の DataMove の説明を指定します。
- ・ Properties("Flags") では、オプションで DataMove 処理を説明する以下のようなフラグを指定します。
 - － \$\$\$BitNoSrcJournal、ジャーナリングされないデータベースをコピーできます。
 - － \$\$\$BitNoBatchMode、バッチ・モードを使用せずにデータ・コピーを実行します。
 - － \$\$\$BitCheckActivate、マッピングの変更を有効にする前に、ユーザ指定のルーチン \$\$\$CheckActivate^ZDATAMOVE() を呼び出して、アプリケーションのステータスを確認します。
- ・ Properties("LogFile") では、オプションで、既定値以外の場合にログ・ファイル名を指定します。

Warnings 配列には、移動するデータが別のネームスペースからもマッピングされている場合にマッピングのリストが含まれます。この配列には、グローバルまたはグローバル範囲の名前が添え字として付きます。

Errors 配列には、競合によってデータの移動が阻止されるマッピングのリストが含まれます。この配列には、グローバルまたはグローバル範囲の名前が添え字として付きます。

例：

```
Set Properties("Flags")= $$$BitCheckActivate
Set Status=##Class(DataMove.API).Generate("TEST",Properties,Warnings,Errors)
```

データベースのジャーナリングの詳細は、“データ整合性ガイド”の“[ジャーナリング](#)”を参照してください。

バッチ・モードでのプロセスの実行の詳細は、“専用のシステム/ツールおよびユーティリティ”の“[プロセス管理](#)”を参照してください。

5.3 DataMove の開始

5.3.1 DataMove.API.StartCopy(Name As %String) As %Status

DataMove コピーを開始します。これにより、実際のデータ移動が処理されます。

引数：

- ・ Name は、DataMove の名前です。

このメソッドによって DataMove の最初のコピーが一連のバックグラウンド・ジョブとして開始され、これらのジョブによって個々の範囲が移動元データベースから移動先データベースにコピーされます。それと同時に、コピー時に変更されるデータの移動先データベースにジャーナル・トランザクションが適用されます。

コピーが実際に始まる前に、StartCopy() によって Validate() および ValidateSizes() が呼び出され、DataMove が検証されます。検証でエラーが返された場合、StartCopy() は検証エラーを返し、DataMove を開始しません。ユーザはエラーを修正してから、StartCopy() を再度呼び出すことができます。

5.4 DataMove の監視

5.4.1 DataMove.API.GetProperties(Name As %String, ByRef Properties as %String) As %Status

DataMove の現在のプロパティを返します。

引数：

- ・ Name は、DataMove の名前です。
- ・ Properties は、以下のような DataMove プロパティの配列です。
 - Properties("ExpandedState") - 状態の展開された外部形式。
 - Properties("JRNMBToApply") - 処理対象のジャーナル・ファイルの現在の MB 数 = 0。
 - Properties("MaxMBPerMin") - DataMove で移動が許可されている 1 分あたりの最大 MB 数。
 - Properties("MBCopied") - コピー済みの MB 数。DataMove の終了時、すべてのデータがコピーされ、Activate() が呼び出される前、コピーされた MB 数は、ソース・データベースのデータが DataMove の実行中に更新されるシステムの **MBToCopy** の MB 数とは一致しない可能性があります。
 - Properties("MBToCopy") - コピー対象のおおよその MB 数。
 - Properties("State") - 移動の現在の状態。以下の表を参照してください。
 - Properties("StateExternal") - **State** プロパティの外部形式。
 - Properties("Status") - 発生したエラーの %Status の値。
 - Properties("Stop") - DataMove の停止状態：
 - ・ \$\$\$\$DMStopNone - 停止のシグナルなし。
 - ・ \$\$\$\$DMStopNormal - Activate() メソッドによって呼び出されたときに、停止のシグナルが送信されました。
 - ・ \$\$\$\$DMStopShutdown - 通常のシステム・シャットダウンにより停止のシグナルが送信されました。システム起動時に再起動するパラメータが設定されている場合、DataMove はシステム再起動時に再開します。
 - ・ \$\$\$\$DMStopUser - ユーザのメソッド StopCopy() の呼び出しにより、停止のシグナルが送信されました。コピーは、StartCopy() によって再開できます。
 - ・ \$\$\$\$DMStopForce - Force() によって停止のシグナルが送信されました。コピーは、StartCopy() によって再開できます。
 - ・ \$\$\$\$DMStopErrorRecoverable - エラーにより停止のシグナルが送信されました。エラーの修正後に、StartCopy() によってコピーを再開できます。
 - ・ \$\$\$\$DMStopErrorUnrecoverable - 回復不能エラーにより停止のシグナルが送信されました。これはジャーナル・エラーなど複数の異なる原因による可能性があります。ここで実行できる唯一の方法は、Rollback() を呼び出してから StartCopy() を呼び出して DataMove を再起動するか、Delete() を呼び出して DataMove を削除することです。

例：

```
%SYS>Set x=##Class(DataMove.API).GetProperties("ABC1",Properties)
```

```
%SYS>Zwrite Properties
Properties("JRNMBToApply")=0
Properties("MaxMBPerMin")=1200
Properties("MBCopied")=17030.67
Properties("MBToCopy")=20034.44
Properties("State")=6
Properties("StateExternal")="Copy"
Properties("Status")=1
Properties("Stop")=0
```

5.5 ^DATAMOVE ユーティリティ

^DATAMOVE ユーティリティは、システムに定義されているすべての DataMove の現在の状態を表示します。ユーザはこれを使用して DataMove の既定のパラメータを設定できます。一度に 1 つの DataMove のみ実行できます。

5.5.1 DataMove の既定のパラメータの設定

DataMove の既定のシステム・パラメータを使用して、システムの再起動時に実行する処理を指定できます。また、1 分あたりに移動するデータの量を制限することもできます。既定で、実行中の DataMove は可能な限り高速にデータをコピーします。ただし、これによりシステム上の他のユーザ・プロセスや、ミラー・システムの遅延に影響が及ぶ可能性があります。

```
%SYS>d ^DATAMOVE
1) Monitor DataMove
2) Edit DataMove settings
3) Exit
Option? 2

Restart DataMove after normal system shutdown? Yes => Yes
Restart DataMove after system crash? Yes => Yes

Setting the maximum number of MB DataMove is allowed to copy per minute
to 0 means that DataMove will copy the data as fast as it can. Changing
this setting will apply the new value to future DataMoves, and optionally
any currently running DataMoves. You can also change this setting through
the 'Metrics' display screen for a running DataMove without changing the
system default.
Maximum number of MB to copy per minute? 0 => 1200
Do you want to apply this change to existing Data Moves? Yes => Yes
Confirm changes? Yes => Yes
DataMove settings updated

1) Monitor DataMove
2) Edit DataMove settings
3) Exit
```

5.5.2 DataMove の監視

Generate() メソッドを呼び出した後は、^DATAMOVE を使用して、移動の状態および移動対象として設定されている範囲を確認できます。

```
%SYS>d ^DATAMOVE
1) Monitor DataMove
2) Edit DataMove settings
3) Exit
Option? 1
```

これは、Generate() および StartCopy() を呼び出した後に表示される内容の例です。これは、DataMove 内のすべての移動対象範囲の累積統計データを示しています。

```

LATEST InterSystems IRIS TRM:19004
File Edit Help

Data Move Monitor
# Name      MBCopied  MBToCopy  Pct  JrnMB  State
1 TEST      410.1    23612.5   2    0      Copy
Status: OK

Last Updated: 2021-02-11 11:26:49.687
Est Finish Time: 2021-02-11 11:46:18.940
Jrn Cycle Time: 0:00:00:00.014
Jrn Cycle End: 2021-02-11 11:26:48.713
Jrn Cycle Start: 2021-02-11 11:26:48.699
First Jrn Time: 2021-02-11 11:26:49.685
First Jrn End: 2021-02-11 11:26:25.428
First Jrn Start: 2021-02-11 11:26:24.382
CopyTime: 2021-02-11 11:26:29.594
End Copy: 2021-02-11 11:26:25.428
Start Copy: 2021-02-11 11:26:24.382
Size Time: 2021-02-11 11:26:25.428
End Size: 2021-02-11 11:26:25.428
Start Size: 2021-02-11 11:26:24.382
Journal Start: :\\latest\\mgr\\journal\\MIRROR-STEVE\\MIRROR-20210211.003
Journal Current: :\\latest\\mgr\\journal\\MIRROR-STEVE\\MIRROR-20210211.003
Done/Avoid/Copy Sets: 0/0/0
Kills: 0/0/0
BitSets: 0/0/0
ZKILLS: 0/0/0
Max MB/Min: 1200
MB/Min: 1201.3
Blks copied: 27951
JRN Count/Size: 0/0
Pid Mov/Jrn: 15768/36064
Pid Copy: 7512
249360
499954404

(R)anges, (J)obs, (M)etrics, (H)alt, (Q)uit =>

```

これらのフィールドは、以下の情報を示しています。

- ・ Jrn Cycle Time - ジャーナル・スキャナまたはジャーナルの適用において、前回ジャーナルの末尾に達したときから、現在のジャーナル・ファイルの末尾に達するまでにかかった時間。
- ・ First Jrn Time - ジャーナル・スキャナまたはジャーナルの適用において、初めて現在のジャーナル・ファイルの末尾に達するまでにかかった時間。
- ・ CopyTime - グローバルのコピーにかかった時間。
- ・ Size Time - グローバルのサイズの計算にかかった時間。
- ・ Done - ジャーナル・ファイル内で検出された各操作のうち、移動先データベースに適用されたものの総数。
- ・ Avoid - ジャーナル・ファイル内で検出された各操作のうち、スキップ可能だったものの総数。コピーがまだ開始していない場合、操作をスキップできます。コピーが開始していた場合は、移動先にまだコピーされていないグローバルの一部で操作が実行されます。
- ・ Copy - コピーの実行中に移動先データベースに適用された各操作の総数。
- ・ Journal Start - DataMove が開始されたジャーナル・ファイルとオフセット。
- ・ Journal Current - 現在のジャーナル・ファイルとオフセット。
- ・ Max MB/Min - DataMove でコピーが許可されている 1 分あたりの最大 MB 数。これを DataMove に設定するには、以下の 3 つの方法のいずれかを使用します。
 - ^DATAMOVE に指定されている DataMove の既定値
 - 以下に示している [メトリック] 画面
 - Modify() API
- ・ MB/Min - DataMove の生存期間中に DataMove がコピーした 1 分あたりの MB 数。
- ・ Blks Copied - コピーされたデータベース・ブロックの実際の数と、事前フェッチされたブロックの割合。

- ・ JRN Count/Size - 対象の DataMove について、現在のジャーナル・ファイル、および StartCopy() が呼び出される前の前回のジャーナル・ファイルで検出されたトランザクションの数と合計サイズ。
- ・ Pid Move/Jrn - コピーおよびデジャーナルを処理する 2 つのプロセスのプロセス ID。
- ・ Pid Copy - マスタ DataMove プロセスのプロセス ID。

範囲オプションを選択した場合、移動対象のグローバルの範囲が 5 つあることがわかります。

#	Name	MBCopied	MBToCopy	Pct	JrnMB	State
1	TEST	917.3	23612.5	4	0	Copy

Status: OK

SrcDB->DstDB	Range	MBCopied	MBToCopy	Pct	JrnMB	State
1 USER->TEST	X(1):(5000001)	58.3	58.3	100	0	JRNApplying
2 USER->TEST	X(400000001):(800000001)	108.5	4664.2	2	0	Copying
3 USER->TEST	X(800000001):(805000001)	58.3	58.3	100	0	JRNApplying
4 USER->TEST	X(1200000001):(1600000001)	140.0	4664.2	3	0	Copying
5 USER->TEST	X(1600000001):(2000000001)	140.0	4664.2	3	0	Copying

(E)xamine, (J)obs, (M)etrics, (H)alt, (Q)uit =>

範囲の詳細画面で、各範囲の詳細を確認できます。これらは上記で説明されていますが、選択した範囲にのみ適用されます。

```

LATEST InterSystems IRIS TRM:19004
File Edit Help

Data Move Monitor
# Name      MBCopied MBToCopy Pct  JrnMB      State
1 TEST      1350.0   23320.9  6    215       Copy
Status: OK

SrcDB->DstDB  Range
2 USER->TEST  X(40000001):(80000001)      Copying
300.0   4664.2  6    225

Last Updated: 2021-02-11 12:53:15.299      Done/Avoid/Copy
Jrn Cycle Time: 0:00:00:00.578 Sets: 0/0/0
Jrn Cycle End: 2021-02-11 12:53:14.224 Kills: 0/0/0
Jrn Cycle Start: 2021-02-11 12:53:13.646 BitSets: 0/0/0
First Jrn Time: ZKILLS: 0/0/0
First Jrn End: MB/Min: 0.2
First Jrn Start: Blks copied: 38400
CopyTime: JRN Count/Size: 0/0
End Copy: Pid Mov/Jrn: 34760/20996
Start Copy: 2021-02-11 12:53:03.572
Size Time: 0:00:00:01.560
End Size: 2021-02-11 12:53:00.327
Start Size: 2021-02-11 12:52:58.767

(N)ext range, (P)rev range, (Q)uit =>

```

ジョブ画面には、どのジョブがどのデータで実行されているかが示されます。

```

LATEST InterSystems IRIS TRM:19004
File Edit Help

Data Move Monitor
# Name      MBCopied MBToCopy Pct  JrnMB      State
1 TEST      1317.4   23612.5  6    0        Copy
Status: OK

# Type Job#  Commands  Globals State  PID Range
1 Copy  65    442110    1347  HANGW    7512 Master Copy
2 Move  71    41242    714  HANG     15768 Master Move
3 Jrn   72    145234    805  LOCKW    36064 Master Journal
4 Copy  77    775743   1815517 HANGW    13232 X(1):(40000001)
5 Scan  74    14374    957  HANGW    31456 X(1):(5000001)
6 Copy  84    849746   1989534 HANG     30860 X(40000001):(80000001)
7 Scan  73    14924    896  HANGW    21816 X(40000001):(80000001)
8 Copy  85    701129   1641305 HANGW    34072 X(80000001):(120000001)
9 Scan  74    14374    957  HANGW    31456 X(80000001):(80500001)
10 Copy 80    818184   1915425 HANGW    31716 X(120000001):(160000001)
11 Scan 76    10799    766  HANGW    10184 X(120000001):(160000001)
12 Copy 82    818277   1915430 HANGW    37936 X(160000001):(200000001)
13 Scan 75    11117    794  HANGW    33048 X(160000001):(200000001)

(R)anges, (M)etrics, (H)alt, (Q)uit =>

```

ここでは、ジョブ 84 が現在、範囲 X(40000000) のデータをコピーしていて、ジョブ 73 が同じ範囲でジャーナルをスキャンしていて、この範囲内で検出したトランザクションを適用していることがわかります。“Master Move” ジョブ 71 は、まだデータをコピー中のすべてのプロセスを制御します。“Master Journal” ジョブ 72 は、ジャーナルをスキャンまたは適用中のすべてのプロセスを制御します。“Master Copy” ジョブは、統計を累積し、Activate() メソッドを呼び出せるタイミン

グを判断します。ここに表示されるジョブ番号、コマンド、グローバル、状態、およびプロセスIDは、JOBEXAMまたは管理ポータルのプロセス表示に示される同じ情報に対応しています。

メトリック画面には、システム上のアクティビティと、実行中の DataMove にそのアクティビティがどの程度関係しているかが示されます。

```

LATEST InterSystems IRIS TRM:19004
File Edit Help

Data Move Monitor

# Name          MBCopied MBToCopy Pct  JrnMB  State
1 TEST          4150.0   23320.9 18   187    Copy
Status: OK

                                Total    Data Move  DM %
Global references/Sec:         977754      962832   98.5
Global update references/Sec:  951904      948668   99.7
Physical reads/Sec:           14167      14167  100.0
Journal Entries/Sec:          952151      948668   99.6
Current MB Copied/Min:                               6248
Max allowed MB Copied/Min:                               Unlimited

Mirror Member Name      Member Type Status    Dejournaling
USP5540STEVEC/1STC      Backup     Active    11 seconds behind
USP5540STEVEC/ASYNC     Async      Async     6 seconds behind

(R)anges, (J)obs, (C)hange rate, (H)alt, (Q)uit =>

```

この例では、DataMove でコピーが許可される 1 分あたりの最大 MB 数が 0 (無制限) に設定されていることがわかります。また、DataMove がグローバル参照とジャーナル・エントリの大部分を占有していて、ミラー処理の遅延が増加する原因になっていることもわかります。

以下のように実行することで、DataMove の処理速度が低減されるよう制限できます。

```

LATEST InterSystems IRIS TRM:19004
File Edit Help

Data Move Monitor
# Name      MBCopied MBToCopy Pct JrnMB      State
1 TEST      7250.0   23320.9 31   1144      Copy
Status: OK

Total      Data Move  DM %
Global references/Sec:      1173847      1154025 98.3
Global update references/Sec: 1141421      1137048 99.6
Physical reads/Sec:         17103         17103 100.0
Journal Entries/Sec:        1141811      1137048 99.6
Current MB Copied/Min:              7020
Max allowed MB Copied/Min:      Unlimited
Mirror Member Name      Member Type Status      Dejournaling
USP5540STEVEC/1STC      Backup      Active      10 seconds behind
USP5540STEVEC/ASYNC      Async      Async      1 second behind

Max allowed MB Copied/Min is currently set to 0-Unlimited
New value 0 => 1200

```

これで、処理速度が更新され、メトリックが大幅に低減されました。

```

LATEST InterSystems IRIS TRM:19004
File Edit Help

Data Move Monitor
# Name      MBCopied MBToCopy Pct JrnMB      State
1 TEST      12781.3  23320.9 55     0         Copy
Status: OK

Total      Data Move  DM %
Global references/Sec:      388243      348239 89.7
Global update references/Sec: 351940      343114 97.5
Physical reads/Sec:         5211         5211 100.0
Journal Entries/Sec:        352694      343114 97.3
Current MB Copied/Min:              1312
Max allowed MB Copied/Min:      1200
Mirror Member Name      Member Type Status      Dejournaling
USP5540STEVEC/1STC      Backup      Active      Caught up
USP5540STEVEC/ASYNC      Async      Async      2 seconds behind

(R)anges, (J)obs, (C)hange rate, (H)alt, (Q)uit =>

```

5.6 マッピング変更の有効化と DataMove の完了

5.6.1 DataMove.API.Activate(Name As %String, Display as %Boolean, Timeout As %Integer = 120) As %Status

DataMove を完了し、ネームスペースのマッピング変更を有効化します。

引数：

- ・ Name は、DataMove の名前です。
- ・ Display は、進捗状況メッセージを確認する場合は 1 に設定する必要があります。
- ・ Timeout は、DataMove が実行を完了し、ジャーナル処理を適用する間、続行を待機する秒数です。

このメソッドにより、DataMove バックグラウンド・ジョブが停止し、ジャーナル・ファイルの処理が完了し、マッピング変更が CPF に書き込まれ、マッピング変更が有効になります。また、他のプロセスが実行を妨げないように、スイッチ 10 が一時的に設定されます。

\$\$\$BitCheckActivate が設定されている場合、続行する前にこのメソッドによりユーザ指定のルーチン \$\$CheckActivate^ZDATAMOVE() が呼び出され、実行されます (存在する場合)。\$\$CheckActivate^ZDATAMOVE() から 1 が返されない場合、メソッドは新しいマッピングを有効にせず終了し、DataMove は実行中のままになります。

注釈 新しいマッピングを有効にするためにスイッチ 10 が設定される直前に、ユーザ指定のルーチンが呼び出されます。このルーチンは、新しいマッピングを有効にする前に、ユーザが希望することをすべて実行できます。

スイッチ 10 の詳細は、“専用のシステム/ツールおよびユーティリティ” の “[スイッチの使用法](#)” を参照してください。

注釈 Activate() により DataMove の State プロパティがチェックされ、最初のコピーが完了し、続行する前にジャーナル適用がキャッチアップされたことが確認されます。

5.7 移動元のグローバルの削除と DataMove の完了

コピーが完了し、更新されたネームスペースが有効化されると、アプリケーション・データが正常にコピーされ、マッピングが有効化されていることを確認する必要があります。これを確認した後は、移動元にある移動済みのグローバルの削除処理に進み、DataMove を完了することができます。

5.7.1 DataMove.API.DeleteSourceGlobals(Name As %String) As %Status

DataMove でコピーされたグローバルを移動元ディレクトリから削除します。

引数：

- ・ Name は、DataMove の名前です。

このメソッドにより、移動先データベースにコピー済みの、移動元データベース内のすべてのグローバルが削除されます。移動元のグローバルを削除するために、複数のプロセスが作成される場合があります。

5.7.2 DataMove.API.Finish(Name As %String) As %Status

DataMove プロセスを完了します。

引数：

- ・ Name は、DataMove の名前です。

このメソッドにより、ログ・ファイルに成功または失敗のメッセージが書き込まれ、ログ・ファイルが閉じられ、DataMove の State プロパティが \$\$\$DMStateDone に設定されます。また、ログ・ファイルがファイル DataMove.log にコピーされます。このファイルは、システムで実行されたすべての DataMove の記録です。

5.7.3 DataMove.API.Delete(Name As %String) As %Status

DataMove プロセスをクリーンアップします。

引数：

- ・ Name は、DataMove の名前です。

このメソッドにより、DataMove が削除され、一時ストレージがクリーンアップされます。

削除したい DataMove 処理が開始されてから停止した場合は、まず Rollback() メソッドを呼び出して、移動されていたデータがあればそれをロールバックします。

5.8 その他の API 呼び出し

5.8.1 DataMove.API.Validate(Name As %String) As %Status

DataMove を検証します。

引数：

- ・ Name は、DataMove の名前です。

DataMove の検証には、指定したマッピングすべての確認、移動元および移動先のデータベースの確認、移動先のグローバルがまだ存在していないことの確認が含まれます。何らかのエラーが発生した場合は Status で報告されます。

Validate() は、StartCopy() の一部として呼び出されます。このメソッドを使用して、StartCopy() でコピーを実際に開始する前に DataMove を検証できます。

5.8.2 DataMove.API.Modify(Name as %String, byref Properties as %String) as %Status

既存のまたは実行中の DataMove に変更を加えます。

引数：

- ・ Name は、DataMove の名前です。
- ・ Properties は、変更するプロパティの配列で、以下が含まれる可能性があります。
 - － Properties("MaxMBPerMin") - DataMove 処理で移動先データベースに移動できる 1 分あたりの最大 MB 数。これを 0 に設定すると、DataMove は可能な限り高速で実行されます。実行中の DataMove に対してこれ呼び出して、処理速度を変更することができます。

5.8.3 DataMove.API.ValidateSizes(Name As %String) As %Status

DataMove によってコピー対象に指定されたデータに対し、十分な容量が存在することを確認します。

引数：

- ・ Name は、DataMove の名前です。

DataMove のサイズの検証には、コピーするデータ量の特定、および移動先データベースに十分な容量があることの確認が含まれます。何らかのエラーが発生した場合は Status で報告されます。

ValidateSizes() は、StartCopy() の一部として呼び出されます。このメソッドを使用して、StartCopy() でコピーを実際に開始する前に DataMove のサイズ要件を検証できます。

5.8.4 DataMove.API.StopCopy(Name As %String) As %Status

DataMove コピー・ジョブを停止します。

引数：

- ・ Name は、DataMove の名前です。

このメソッドにより、DataMove コピー・バックグラウンド・ジョブが停止し、進行中のコピーを適切に停止できるようになります。DataMove の再起動は、StartCopy() を使用して行えます。

5.8.5 DataMove.API.Rollback(Name As %String) As %Status

DataMove をロールバックします。

引数：

- ・ Name は、DataMove の名前です。

このメソッドにより、DataMove コピー・ジョブによって移動先データベースにコピーされたグローバルがすべて削除されます。このメソッドは、DataMove の中止またはコピー・プロセスでのエラーからの回復に使用できます。

このメソッドを呼び出す前に StopCopy() を実行する必要があります。

ロールバックの実行後は、StartCopy() を使用してやり直すことも、Delete() を使用して Data Move を削除することもできます。

5.8.6 DataMove.API.RollbackMappings(Name As %String)

DataMove マッピングをロールバックします。

引数：

- ・ Name は、DataMove の名前です。

これにより、システムのマッピングが DataMove Name の実行前の状態に復元されます。DataMove Name の State が \$\$\$DMStateNSPActivateDone である必要があります。この使用例として、DataMove でマッピングが有効化された後にアプリケーションのテストを開始し、何らかの問題が検出された場合が考えられます。マッピングが以前の値に復元されると、State は \$\$\$DMStateJrnApplyDone に設定されます。その後で、Rollback() を呼び出すか (推奨)、StartCopy() を呼び出して有効化を再試行することができます。

5.8.7 DataMove.API.RollbackCopy(Name As %String, Warnings as %String, Errors as %String)

DataMove のコピーをロールバックします。

引数：

- ・ Name は、DataMove の名前です。
- ・ Warnings は、DataMove の実行を阻止することはない競合を含んで返される配列です。
- ・ Errors は、DataMove の実行を阻止する競合を含んで返される配列です。

これにより、Name-ROLLBACK という名前の DataMove が作成されます。この DataMove を実行すると、DataMove Name でコピーされたデータが、元の移動元ディレクトリに戻されます。DataMove Name の State は \$\$\$DMStateDeleteSrcGlobalsDone である必要があります (移動元のグローバルはすべて DeleteSourceGlobals() メソッドによって削除されています)。このメソッドが完了すると、StartCopy() および Activate() を実行して、データを元のディレクトリに戻すことができます。これにより、DataMove Name の State は \$\$\$DMStateRollbackCopy に設定されます。

5.8.8 Query DataMove.API.ListDMs(Names As %String, Flags as %Integer=0) As %Status

すべての DataMove およびそのプロパティのリストを返すクエリ。

引数：

- ・ Names は、DataMove のコンマ区切りのリストです。
- ・ フラグ -

- 0 = DataMove レコードに記録されている現在の情報を返します。
- 1 = すべての範囲をまとめた情報を返します。

5.8.9 Query DataMove.API.ListRanges(Name As %String, SrcDBs As %String = "*", DstDBs As %String = "*", Ranges As %String = "*", Flags = 0) As %Status

すべての DataMove 範囲およびそのプロパティのリストを返すクエリ。

引数：

- ・ Names は、DataMove の名前です。
- ・ SrcDBs - 移動元データベースのコンマ区切りのリスト
- ・ DstDBs - 移動先データベースのコンマ区切りのリスト
- ・ Ranges - 範囲のコンマ区切りのリスト
- ・ フラグ -
 - 0 = フォーマットされた時刻を返します。
 - 1 = \$h UTC 時刻として時刻を返します。

5.8.10 Query DataMove.API.ListProcesses(Name As %String) As %Status

すべての DataMove プロセスのリストを返すクエリ。

引数：

- ・ Name は、DataMove の名前です。

6 DataMove の状態

DataMove では、State プロパティを使用して、ワークフロー全体を通して進捗状況が記録されます。GetProperties() を使用することで、このプロパティを調べて、進捗状況を監視できます。問題が生じた場合は、クエリ ListDMs() を使用して、トラブルシューティングすることができます。

各 State の整数値は、%syDataMove.inc インクルード・ファイルで定義されています。

テーブル 1: DataMove の State のリスト

State	説明
\$\$\$DMStateNotStarted	DataMove を作成するために Generate() が呼び出されました。
\$\$\$DMStateStarted	コピーを開始するために StartCopy() が呼び出されました。
\$\$\$DMStateSize	移動するデータのサイズが計算されています。
\$\$\$DMStateSizeDone	移動するデータのサイズが計算されました。
\$\$\$DMStateCopy	ソース・データが移動先データベースにコピーされています。

State	説明
\$\$\$DMStateCopyDone	ソース・データの移動先データベースへのコピーが完了しました。
\$\$\$DMStateJrnApply	すべてのソース・データが移動先データベースにコピーされ、現在ジャーナルが移動先データベースに適用されています。Activate() または StopCopy() が呼び出されるまで、またはシステムが正常にシャットダウンするまで、この状態が維持されます。
\$\$\$DMStateJrnApplyDone	Activate() または StopCopy() の呼び出し中に、DataMove プロセスが停止しました。
\$\$\$DMStateNSPActivate	Activate() が呼び出されました。スイッチ 10 が設定され、新しいネームスペース・マッピングが有効化されています。
\$\$\$DMStateNSPActivateDone	新しいネームスペース・マッピングが有効化されました。
\$\$\$DMStateDeleteSrcGlobals	DeleteSourceGlobals() が呼び出され、移動元のグローバルが削除されています。
\$\$\$DMStateDeleteSrcGlobalsDone	移動元のグローバルがすべて削除されました。
\$\$\$DMStateDone	Finish() が呼び出されました。すべてのデータが移動先データベースに移動され、ネームスペースが有効化され、最終的なログ・ファイルが更新されました。
\$\$\$DMStateRollback	Rollback() メソッドが呼び出され、コピーされたすべての移動先グローバルが削除されています。完了すると、State は \$\$\$DMStateNotStarted に設定されます。
\$\$\$DMStateRollbackCopy	RollbackCopy() メソッドが呼び出されました。この DataMove は完了し、ここからデータを前の場所に戻す新しい DataMove が作成されました。

7 DataMove の例

ObjectScript ルーチンは、DataMover という DataMove オブジェクトを使用して、ネームスペース LIVE のグローバル ABC および DEF をデータベース LIVE-CT に移動します。

この例は、デモンストレーションを目的としているため、包括的なエラー・チェックは含まれていません。各 API 呼び出しの後に返されるステータスを確認してください。

```
#include %occStatus
#include %syDataMove
#include %syConfig

#; Sample routine to illustrate using DataMove to move globals.
Example() {
    #; New Mapping definition
    Set NewMap = 1
    Set NewMap(1, "name") = "ABC"
    Set NewMap(1, "db") = "LIVE-CT"
    Set NewMap = 2
    Set NewMap(2, "name") = "DEF"
    Set NewMap(2, "db") = "LIVE-CT"

    #; Let's start initializing the edits space before moving the globals
    Write "Initialize edits....."
```

```

Set Namespace = "LIVE"
Set Status = ##class(DataMove.API).MapInitialize("DBNAME",Namespace)

#; Now we need to create the new mappings
Write "Create/Update maps....."

#; Main loop to define mappings and operations to be done.
For i = 1:1:NewMap {
    Kill Obj, Name, Prop, Status
    Set Prop("Database")=NewMap(i, "db")
    Set Name=NewMap(i, "name")

    #; Write name after a new line
    Write !, Name

    #; Checking if the global exists
    Do ##Class(Config.MapGlobals).Exists(Namespace, Name, .Obj)
    If (Obj = "") {
        #; Creating the global
        Write !, " - Create....."
        Set Status = ##class(DataMove.API).MapGlobalsCreate("DBNAME",Namespace,Name,.Prop)
    } Else {
        #; Modifying the global
        Write !, " - Modify....."
        Set Status = ##class(DataMove.API).MapGlobalsModify("DBNAME",Namespace,Name,.Prop)
    }
}

#; We have everything set up, now we need to create the DM object
Write "Create DM....."
Set Status = ##class(DataMove.API).Generate("DBNAME",.Properties, .Warnings, .Errors)
If $data(Warnings) Write ! Zwrite Warnings
If $data(Errors) Write ! Zwrite Errors

#; Creating the background job that needs to be run
Write "Creating the JOB....."
Set Status = ##class(DataMove.API).StartCopy("DBNAME")

#; Monitor the DataMove until we can activate it.
For {
    #; Monitor state every 5 seconds
    Hang 5
    Set Status=##Class(DataMove.API).GetProperties("DBNAME",.Properties)
    #; Activate will only run if we are applying journals and have 5 MB or less of journal to apply.

    If (Properties("State")=$$$DMStateJrnApply) {
        If (Properties("JRNMBToApply")<=5) {
            Quit
        }
    }
    Write !,"MB to copy: "_Properties("MBToCopy")
    Write !,"MB copied: "_Properties("MBCopied")
    Write !,"Journal MB still to apply: "_Properties("JRNMBToApply")
}
Write !, "Activating the new Mappings..."
Set Display=1
Set Status=##Class(DataMove.API).Activate("DBNAME",Display)

#; Delete the source globals that have been copied in the DataMove
Write "Deleting the old Globals....."
Set Status = ##class(DataMove.API).DeleteSourceGlobals("DBNAME")

#; Stopping the DataMover
Write "Closing the DataMover....."
Set Status = ##class(DataMove.API).Finish("DBNAME")

#; Delete the DataMove and clean up any temporary storage
Write "Deleting the DataMover....."
Set Status = ##class(DataMove.API).Delete("DBNAME")
}

```

8 非推奨の呼び出しと State

呼び出しと State の両方に変更が加えられています。以下の 2 つの表は、古い非推奨の呼び出しと State、およびそれに関連付けられている代替の呼び出しと State をリストしています。

テーブル 2: 非推奨および新規の DataMove の呼び出しのリスト

古い名前	新しい名前
Config.Map.InitializeEdits	DataMove.API.MapInitialize
Config.MapGlobals.Create	DataMove.API.MapGlobalsCreate
Config.MapGlobals.Delete	DataMove.API.MapGlobalsDelete
Config.MapGlobals.Modify	DataMove.API.MapGlobalsModify
Config.Map.Revert(0)	DataMove.API.RollbackCopy
Config.Map.Revert(1)	DataMove.API.RollbackMappings
DataMove.Data.Activate	DataMove.API.Activate
DataMove.Data.CreateFromMapEdits	DataMove.API.Generate
DataMove.Data.Dispatch(Name, "Delete")	DataMove.API.Delete
DataMove.Data.Dispatch(Name, "DeleteSrcGlobals")	DataMove.API.DeleteSourceGlobals
DataMove.Data.Dispatch(Name, "Finish")	DataMove.API.Finish
DataMove.Data.Dispatch(Name, "Rollback")	DataMove.API.RollBack
DataMove.Data.Dispatch(Name, "Validate")	DataMove.API.Validate
DataMove.Data.Dispatch(Name, "ValidateSizes")	DataMove.API.ValidateSizes
DataMove.Data.DispatchJob(Name, "Copy")	DataMove.API.StartCopy
DataMove.Data.Display()	^DATAMOVE
DataMove.Data.Exists	代替の呼び出しなし
DataMove.Data.StopCopy	DataMove.API.StopCopy

テーブル 3: 非推奨および新規の DataMove の State のリスト

古い State	新しい State
\$\$\$Copy	\$\$\$DMStateCopy
\$\$\$CopyDone	\$\$\$DMStateCopyDone
\$\$\$CPFUpdate	対応する State なし
\$\$\$CPFUpdateDone	対応する State なし
\$\$\$Delete	\$\$\$DMStateDeleteSrcGlobals
\$\$\$Done	\$\$\$DMStateDone
\$\$\$JrnApply	\$\$\$DMStateJrnApply
\$\$\$JrnApplyDone	\$\$\$DMStateJrnApplyDone
\$\$\$NotStarted	\$\$\$DMStateNotStarted
\$\$\$NSPActivate	\$\$\$DMStateNSPActivate
\$\$\$NSPActivateDone	\$\$\$DMStateNSPActivateDone
\$\$\$Size	\$\$\$DMStateSize

古い State	新しい State
\$\$\$SizeDone	\$\$\$DMStateSizeDone
\$\$\$Started	\$\$\$DMStateStarted
対応する State なし	\$\$\$DMStateRollback
対応する State なし	\$\$\$DMStateRollbackCopy